

Week 2 – Heroes and Cowards Tutorial

CSYS5010 – INTRODUCTION TO COMPLEX SYSTEMS

1. Launch NetLogo.
2. In the Interface tab open the world settings. Untick the horizontal and vertical wrap tickboxes. This creates the room walls.
3. At the top of the code tab add the following command

```
turtles-own [ friend enemy ]
```

This command declares two variables that belong to turtles: friend and enemy. We will use these variable to store references to other turtles.

4. Create the setup procedure with the following code

```
to setup
  clear-all
  ask patches [set pcolor white ] ;; create a blank background
  create-turtles number [
    setxy random-xcor random-ycor
    ;; set the turtle personalities based on chooser
    if (personalities = "heroes") [ set color blue ]
    if (personalities = "cowards") [ set color red ]
    ;; choose friend and enemy targets
    set friend one-of other turtles
    set enemy one-of other turtles
  ]
  reset-ticks
end
```

First the setup procedure clears the screen and sets the color of patches to white. Then follows a long command `create-turtles number [...]`. The command `create-turtles` creates a number of turtles and then immediately provides them with some instructions in the square brackets. We will use the global variable `number` to designate the number of turtles we start with. We will control the variable with a slider in the interface, which we will create later. The command `setxy random-xcor random-ycor`

sets the coordinate of each turtle to a random value. The commands

```
if (personalities = "heroes") [ set color blue ]
if (personalities = "cowards") [ set color red ]
```

set the color of the turtle depending if the game set to “coward” or “heroes” mode. These commands query the global variable `personalities`, which we will declare later.

```
set friend one-of other turtles
set enemy one-of other turtles
```

These commands tell each turtle to choose another turtle from the set of all turtles and set the `FRIEND` variable to take on the value of that turtle. Thus `FRIEND` and `ENEMY` variables will store references to turtles.

5. Go to the interface tab.
6. Add the setup button and link it to the setup procedure.
7. Create a slider and write number in the Global Variable field. This declares the global variable `number` that we use to control the number of created turtles.
8. Create a chooser and denote write personalities in the Global Variable field. As possible options write “heroes” and “cowards”.
9. Test the setup procedure.




Figure 1: Two possible Initial conditions with 50 agents in heroes and cowards game mode

10. In the Code tab create a go procedure by adding the following code

```
to go
  ask turtles [
    if (color = blue) [ act-bravely ]
    if (color = red) [ act-cowardly ]
  ]
  tick
end
```

This code asks turtles to act bravely or act cowardly depending on their color. We now need to create the procedures act-bravely and act-cowardly.

11. Add the following code to create the act-bravely procedure

```
to act-bravely
  ;; move toward the midpoint of your friend and enemy
  facexy ([xcor] of friend + [xcor] of enemy) / 2
  ([ycor] of friend + [ycor] of enemy) / 2
  fd 0.1
end
```

Add the following code to create the act-cowardly procedure

```
to act-cowardly
  ;; put your friend between you and your enemy
  facexy [ xcor] of friend + ([xcor] of friend - [ xcor] of enemy) / 2
  [ ycor] of friend + ([ycor] of friend - [ ycor] of enemy) / 2
  fd 0.1
end
```

12. In the Interface create the go button. Run the simulation. What kind of behaviour do you observe?
13. Load the ready-made Heroes and Cowards model from the model library, located in IAMB Textbook | Chapter 2 | Heroes and Cowards.nlogo.
14. Investigate the model. Notice the addition of the “mixed” personality – what does it do? Observe the sensitivity of the game to initial conditions and emergence of different types of dynamics.




Figure 2: Heroes and Cowards model from the model library, which contains mixed personalities mode.