


Week 2 – Simple Economy Tutorial

CSYS5010 – INTRODUCTION TO COMPLEX SYSTEMS

1. Launch NetLogo.
2. In the Code tab add the following line to declare the wealth variable that denotes the wealth of a turtle
`turtles-own [wealth]`
3. In the Interface tab create a plot widget. Name it wealth distribution and input the following commands into the Pen update command prompt accessible after clicking the  button. Set the mode to Bar and the Interval to 5.0
`set-plot-y-range 0 40`
`histogram [wealth] of turtles`
This command set the y range of the plot and instructs the plot to become a histogram of the wealth of the turtles.
4. Adjust the scale of the histogram by setting X max to 500 , Y max to 40 and uncheck the Auto scale? Tickbox.
5. Adjust the World setting by changing the location of origin to “corner”, the max-pxcor to 500, max-pycor to 80 and Patch size to 1.
6. Go to the Code tab and create the setup procedure. Here is the code for the setup procedure

```
to setup
  clear-all
  create-turtles 500 [
    set wealth 100
    set shape "circle"
    set color green
    set size 2
    ;; visualize the turtles from left to right in ascending order of wealth
    setxy wealth random-ycor
  ]
  reset-ticks
end
```

This code clears the world, creates 500 turtle agents passing several instructions to them and finally resets clock. The instructions to each turtle is to set the value of its wealth to 100, change its appearance to a green circle of size 2 and finally set its x coordinate to the value of its wealth and y coordinate to a random value.

7. Test the setup procedure.




Figure 1: Simple economy - testing the setup procedure.

8. Create a go procedure by writing using the following code.

```

to go
  ;; transact and then update your location
  ask turtles with [wealth > 0] [transact]
  ask turtles [set xcor wealth ]
  tick
end

```

Here, the first instruction of the procedure is to ask turtles that have non-zero amount of wealth to perform the transact procedure. The transact procedure, which we will create shortly, will encode the agent-to-agent interactions – wealth transactions between the agents. The second instruction sets the x-coordinate of turtles to the value of their wealth.

9. Create the transact procedure using the following code

```

to transact
  ;; give a dollar to another turtle
  set wealth wealth - 1
  ask one-of other turtles [set wealth wealth + 1]
end

```

The transact procedure reduces the wealth of turtle by 1 and increases the wealth of a randomly selected other turtle by 1.

10. Create the go button and test the simulation. Run the simulation for 10000 ticks. What kind of distribution emerges? Is this what you have expected?
11. Finally load the ready made Simple Economy mode from the model library, located at IABM Textbook | Chapter 2 | Simple Economy.nlogo. This model has additional widgets for visualizing the wealth of the top 10% and bottom 50%. Can you figure out how these were created? (Note: reporters/monitors will be covered in following lectures.)